# Why am I having trouble calculating this ODE?

I'm trying to compute solutions for $$\frac{\partial^{4}}{\partial t^{4}}\mathcal{F}_{x}u(\xi,t)-|\xi|^{4}\mathcal{F}_{x}u(\xi,t)=0$$

DSolve[FourierTransform[Derivative[4][u][t], t, ω] -
Abs[ω]^4*FourierTransform[u[t], t, ω] = 0, u[t], t]


But this doesn't seem to work out. Would someone be able to tell me how to properly do this so that I get something which looks like:

$$\mathcal{F}_{x}u(\xi,t)=c_{1}(\xi)e^{\cdots}+c_{2}(\xi)e^{\cdots}$$

Edit: I also tried:

Solve[LaplaceTransform[Derivative[4][u][t], t, ω] -
Abs[ω]^4*LaplaceTransform[u[t], t, ω] == 0, u[t], t]


with no success.

I even cannot solve the equation $$\frac{\partial^{4}}{\partial t^{4}}u(\xi,t)-|\xi|^{4}u(\xi,t)=0$$

DSolve[Derivative[4][u][t] - Abs[ω]^4*u[t] == 0, u[t], t]

• You have =; instead it should be ==. – corey979 Nov 8 '16 at 19:01
• @corey979 Thanks, but I still have problems even having fixed that. – Jason Born Nov 8 '16 at 19:05
• DSolve is for solving differential equations. Once you've taken the FourierTransform of the differential equation, you have an algebraic equation, which means that you can't use DSolve; you have to use Solve. Then, once you've solved for the Fourier transform of u[t], you have to take the inverse Fourier transform, and you're done. However, if this is an initial value problem, I think you have to use either Laplace transforms or Fourier sine and cosine transforms instead. I don't think Fourier transforms are useful for solving IVPs. – march Nov 8 '16 at 19:09
• @march See edit. – Jason Born Nov 8 '16 at 19:17
• What exactly do you want to archieve? Because your ODE is solved perfectly by mathematica with DSolve[D[u[\[Xi],t],{t,4}]-Abs[\[Xi]]^4*u[\[Xi],t]==0,u[\[Xi],t],t]. – Julien Kluge Nov 8 '16 at 19:39

Okay i'm going to show how to solve your ODE in two ways.

1.: Mathematica's DSolve. Lets define arbitrary starting values (not nessecary. Just to compare it with the second method) and our equation:

startValues={u[ξ,0]==1,(u^(0,1))[ξ,0]==1,(u^(0,2))[ξ,0]==0,(u^(0,3))[ξ,0]==0};
equation = D[u[ξ, t], {t, 4}] - Abs[ξ]^4*u[ξ, t] == 0;


Then we can simply plug it into DSolve with a Simplification afterwards (which is also not nessecary but makes the solution a lot more nice to look at):

FullSimplify[DSolve[{equation}~Join~startValues,u[ξ,t],t],{ξ∈Reals,ξ!=0}]


{{u[ξ,t]->(ξ Cos[t ξ]+ξ Cosh[t ξ]+Sin[t ξ]+Sinh[t ξ])/(2 ξ)}}

We use LaplaceTransform[D[u[ξ, t], {t, 4}] - Abs[ξ]^4*u[ξ, t] == 0, t, ω] and plugin this into Solve (since the ODE is now a algebraic expression)

Solve[LaplaceTransform[D[u[ξ,t],{t,4}]-Abs[ξ]^4*u[ξ,t]==0,t,ω],LaplaceTransform[u[ξ,t],t,ω]]


We can now put this into InverseLaplaceTransform:

InverseLaplaceTransform[Solve[LaplaceTransform[D[u[ξ,t],{t,4}]-Abs[ξ]^4*u[ξ,t]==0,t,ω],LaplaceTransform[u[ξ,t],t,ω]],ω,t]


Simplifiyng afterwards and adding our startvalues:

startValueRules = startValues /. Equal -> Rule;
FullSimplify[
InverseLaplaceTransform[
Solve[LaplaceTransform[
D[u[ξ, t], {t, 4}] - Abs[ξ]^4*u[ξ, t] == 0,
t, ω],
LaplaceTransform[u[ξ, t], t, ω]], ω, t] /.
startValueRules, {ξ ∈ Reals, ξ != 0}]


{{u[ξ,t]->(ξ Cos[t ξ]+ξ Cosh[t ξ]+Sin[t ξ]+Sinh[t ξ])/(2 ξ)}}

Which is the same solution as above.

Plotting the solution from above (Attention: your ξ parameter is very volatile for $\xi\neq 1$)

Plot[(ξ Cos[t ξ]+ξ Cosh[t ξ]+Sin[t ξ]+Sinh[t ξ])/(2 ξ)/.ξ->1,{t,-15,2}]